PEGASUS BASIC CHARACTER CREATOR (requires Monitor 2 or later version)

The programmable character RAM option permits the user to define and use an alternative character set to that stored in the character ROM. This BASIC program allows the user to examine the shape of any programmable character and to modify and store it.

When the machine is powered up, the programmable character RAM will contain random bits. When in BASIC, CTRL G will turn on these characters; CTRL C will turn them off. Try this after power up to see how the random patterns appear.

There are a total of 128 characters (with ASCII numbers from 0 to 127). Each character is 8 bits wide and 16 bytes high. The CHARACTER CREATOR program will display a grid 8 wide, and 16 lines high, which you can fill with white blocks, or blanks.

Using the Program

1) Enter the program from the listing and save it on cassette tape. Note lines 230 and 310 with inverse blanks.

2) Type RUN The screen will clear and CHARACTER NUMBER will appear requesting an ASCII number

3) Type in a number from 0 to 127 (press RETURN)

4) WAIT will appear as the present shape of the character is retrieved from RAM

5) The screen will display the ASCII character number, the ROM form of the character in inverse video, and an 8 by 16 grid of [ ] brackets showing the present programmable character shape.

6) If you wish to erase the present shape, press CTRL and DELETE simultaneously and the grid pattern will clear.

7) The cursor will appear in the top left hand block. Use these keys to move the cursor.

   S - left E - up X - down D - right

   When the cursor is placed at a [ ] you wish to fill, press the ? key. To clear the [ ] press the ; key (do not SHIFT)
   By moving the cursor to each bracket block you want filled you can create your own pattern.

8) When that character is defined, press RETURN.
   WAIT will briefly appear as the new character is stored into graphics RAM.

9) NEXT? will appear requesting another ASCII number.
   Use 999 to exit the program

10) These new characters will remain in the graphics RAM as long as power is supplied. To save your characters for later use, use the CHAR SAVE/LOAD program.

```
100 REM CHARACTER CREATOR
110 SIG:CLS:PRINT[0,5],
120 INPUT "CHARACTER NUMBER " N:IF N>127 THEN END
130 CLS:LINES 1:PRINT[0,2],"CHAR ",N
140 PRINT[0,3],CHR(1)," ",RAW(N)," ",CHR(2),
150 GOSUB 1000
200 PRINT CHR(25),CHR(N),:REM SHOW CHAR
210 FOR Y=0 TO 15:Q=PEEK($BDA0+Y):W=128
220 FOR X=0 TO 14 STEP2
230 B=Q-W:IF B>=0 THEN PRINT[14+X,Y],"  ",:Q=B
240 W=W/2:NEXT X:NEXT Y
250 PRINT[0,0],"    ",:LINES 16
260 PRINT[14,0],
300 M=INKEY: IF M=0 THEN 300:REM FORM CHAR
310 IF M=46 THEN PRINT "  ",CHR(19),CHR(19),:GOTO300
320 IF M=44 THEN PRINT "[]",CHR(19),CHR(19),:GOTO300
330 IF M=88 THEN PRINT CHR(24),:GOTO300
340 IF M=69 THEN PRINT CHR(5),:GOTO300
350 IF M=68 THEN IF PEEK($BDF9)<27 THEN PRINT CHR(4),CHR(4),:GOTO300
360 IF M=83 THEN IF PEEK($BDF9)>15 THEN PRINT CHR(19),CHR(19),:GOTO300
370 IF M=13 THEN 400
380 IF M=31 THEN GOSUB 1010
390 GOTO300
400 PRINT[0,0],"WAIT",:LINES 1:REM STORE CHAR
410 FOR Y=0 TO 15:B=0:W=128
420 FOR X=0 TO 14 STEP2
430 V=PEEK($BE0E+X+Y*32):IF V=32 THEN B=B+W
440 W=W/2:NEXT X:@(Y)=B:NEXT Y
450 PRINT CHR(27),CHR(N),
460 FOR Y=0 TO 15:PRINT CHR(@(Y)),:NEXT Y
470 PRINT [0,0],"    ",:LINES 16:PRINT[0,8],
480 INPUT "NEXT" N:IF N>127 THEN END
490 GOTO 130
500 END
1000 PRINT[0,0],"WAIT",:REM GRID
1010 FOR Y=0 TO 15
1020 PRINT[10,Y],Y+1,[14,Y],"[][][][][][][][]",
1030 NEXT Y:PRINT[14,0],:RETURN
```

LINE 230 - after the first " press the inverse video key (far lower right), press space bar twice, press inverse key again, and resume with the second "

LINE 310 - do the same as above

Suggestions

Define NUMBER 32 as blank - this is the same as a space in the character ROM. Define any other character(s). To see how they look in normal size, type CTRL G to change to programmable characters, then type the key for the new character.

Return to ROM characters with CTRL C

Press RETURN

ERR 4 will appear. This may be ignored

To display ASCII characters 0 to 31, the RAWMODE key (second blank key from right at lower right) should be pressed.

Then -

CTRL A will display ASCII 1

CTRL B will display ASCII 2

etc.

Press RAWMODE again to return to normal keyboard functions.

Program Comments

LINES	FUNCTION
110 - 140	- input ASCII number for a character - print the number, and display in inverse video the present ROM character shape if in normal characters, or the RAM character shape if in programmable graphics
200	- CHR(25) prepares to read the graphics RAM - CHR(N) specifies the character to be read
210	- the 16 bytes for a character are placed into RAM starting at $ BDA0
220 - 240	- each byte is converted into a bit pattern, printing a white block if the bit is a one. The display is reduced to 1 line for increased computation speed and then returned to 16 lines
300 - 390	- INKEY returns a zero if no key is pressed. Eight commands are recognised.

M = 46	. key fill with white block
M = 44	, key fill with [ ]
M = 88	X key move down
M = 69	E key move up
M = 68	D key move right
M = 83	S key move left
M = 13	RETURN store character
M = 31	CTRL DELETE redisplay empty grid

LINES	FUNCTION
410 - 440	- converts the character pattern to 16 bytes and stores them in the @ array
450	- CHR(27) prepares to write to the graphics RAM - CHR(N) specifies the character to be written
460	- the 16 bytes are written to graphics RAM
480	- the next ASCII number is requested
1000 - 1030	- a grid of [ ] 8 wide by 16 high is displayed